


EM302 Yöneylem Arařtırması 2

Dr. Özgür Kabak

GAMS Giriş

- ▶ GAMS (The General Algebraic Modeling System) matematiksel proglamlama ve optimizasyon için tasarlanan yüksek seviyeli bir dildir.



GAMS Temelleri

- ▶ Kümeler
- ▶ Veri Girişi
- ▶ Değişkenler
- ▶ Denklemler
- ▶ Model
- ▶ Çözüm



Örnek Problem

Sabit maliyetli ulařtırma modeli

- ▶ ATK- Beyaz dört müşterisinin taleplerini karşılamak için üç potansiyel fabrikayı değerlendirmektedir. Her hangi bir fabrikayı açarsa firma 200,000TL'lik yatırım yapması gerekmektedir.
- ▶ Bir fabrikada üretilip, bir müşteriye gönderme birim maliyetleri tabloda verilmiştir.
- ▶ Firmanın talepleri en düşük maliyetle karşılaması için gerekli TP modelini kurunuz.

	Müşteri 1	Müşteri 2	Müşteri 3	Müşteri 4	Kapasite
Fabrika 1	8TL	6TL	10TL	9TL	60,000
Fabrika 2	9TL	12TL	13TL	7TL	55,000
Fabrika 3	14TL	9TL	16TL	5TL	45,000
Talepler	40,000	30,000	25,000	15,000	

Sabit maliyetli ulařtırma modeli - TP

▶ Karar deęiřkenleri

▶ x_{ij} : i fabrikasından j müşterisine gönderilen miktar ($i = 1,2,3; j = 1,2,3,4$)

▶ y_i : i fabrikasının açılması ($i = 1,2,3$)

$$y_i = \begin{cases} 1 & i \text{ fabrikası açılırsa} \\ 0 & \text{açılmazsa} \end{cases}$$

▶ TP modeli

$$\text{Min } \sum_i \sum_j c_{ij} x_{ij} + 200000 \sum_i y_i$$

Öyle ki; $\sum_j x_{ij} \leq k_i y_i$ ($i = 1,2,\dots,m$) kapasite kısıtları

$\sum_i x_{ij} \geq t_j$ ($j = 1,2,\dots,n$) Talep kısıtları

$$x_{ij} \geq 0, y_i \in \begin{Bmatrix} 0 \\ 1 \end{Bmatrix}$$



Kümeler

- ▶ GAMS kapsamında kümeler, cebirdeki alt indislere veya belirli öğeleri içeren dizilere denk gelirler (örnek: fabrikalar, müşteriler...)
- ▶ Kümelerin tanımlanması

SET	kümeadı	açıklayıcı yazı(isteğe bağlı)
/	kümenin_ilk_öğesinin_adı	açıklayıcı yazı(isteğe bağlı)
	kümenin_ikinci_öğesinin_adı	açıklayıcı yazı(isteğe bağlı) /;

```
sets i fabrikalar /1, 2, 3/  
     j müşteriler /1*4/ ;
```



Kümeler

- ▶ Yıldız imi '*', eğer bir küme sıra halinde elemanlar içeriyorsa yazımı kolaylaştırmak için kullanılır

```
SETS
```

```
m makinalar / mak1*mak4/  
i isler      / is1*is4/;
```

Her iki gösterimde aynı kümeyi tanımlar

```
SETS
```

```
m makinalar / mak1, mak2, mak3, mak4/  
i isler      / is1, is2, is3, is4/;
```

- ▶ **alias** komutu ise bir kümenin elemanlarını farklı bir isimdeki başka bir kümeye kopyalamak için kullanılır

```
set
```

```
i /1*5/  
alias (i,j):
```

Her iki küme de (i,j) aynı elemanlardan oluşmaktadır(1,2,3,4,5)



Veri Girişİ

- ▶ GAMS çeşitli yöntemlerle veri girişine olanak sağlar, bunlardan bazıları: scalar, parametre, tablo

Scalar(Sabit)

scalar ifadesi sıfır boyutlu bir katsayı tanımlamak için kullanılır

SCALAR

ilköğeadı	açıklayıcı yazı(isteğe bağlı)	/sayısal değer/
ikinciöğeadı	açıklayıcı yazı(isteğe bağlı)	/sayısal değer/ ;

```
Scalars  
  indo "indirim oranı" /0.15/  
  ivo "iç verim oranı" /0.07/;
```

```
scalar fa /200000/;
```


Veri Girişi

Parameterler

Parametre formatı bir kümenin elemanlarına ilişkin değerler atamak için kullanılır

PARAMETER

ögeadı(bağlıolduğuküme) açıklayıcı yazı(isteğe bağlı)

/kümenin 1. elemanın adı ilgili değer,

kümenin 2. elemanın adı ilgili değer,

..... ;

```
parameter talep(j) /1 40000, 2 30000, 3 25000, 4 15000 /;  
parameter kapasite(i) / 1 60000, 2 55000, 3 45000 /;
```



Veri Girişi

Tablolar

Tablo formatı veri değerleri iki veya daha fazla kümeye dayanan bilgileri girmek için kullanılır

TABLE öğeadı(1.küme, 2.küme) açıklayıcı yazı(isteğe bağlı)
küme2eleman1 küme2eleman2
küme1eleman1 değer_11 değer_12
küme1eleman2 değer_21 değer_22 ;

```
table c(i,j)
  1   2   3   4
1 8   6  10  9
2 9   12 13  7
3 14  9  16  5 ;
```



Değişkenler

- ▶ GAMS içindeki bir değişken optimize edilecek modelde veya çözümü aranan denklemlerde üzerinde oynanabilen sayısal bir büyüklük olarak tanımlanır
- ▶ Değişkenin tanımlanması

Değişken türü

birincideğişkeninadı(ilışkiliküme)

ikincideğişkeninadı(ilışkiliküme)

açıklayıcı yazı(isteğe bağlı)

açıklayıcı yazı(isteğe bağlı) ;

```
positive variables x(i,j);  
binary variables y(i);  
variable z;
```

VARIABLES

```
x(i)      tamsayi karar degiskeni  
y(m)      ikili karar degiskeni  
z         amac fonksiyonu ;
```

```
BINARY VARIABLE      y ;  
INTEGER VARIABLE     x ;
```

Değişkenler

Değişken türü	Değişken Tanımı
Variable	Kısıt yoktur. Değişken eksi sonsundan artı sonsuza kadar bütün değerleri alabilir
Positive Variable	Negatif olmayan sayılar. Değişken sıfırdan artı sonsuza kadar bütün değerleri alabilir
Nonnegative variable	Sadece pozitif değerler. Değişken sıfırdan artı sonsuza kadar bütün değerleri alabilir
Negative Variable	Sadace negatif değerler. Değişken eksi sonsuzdan sıfıra kadar bütün değerleri alabilir
Binary Variable	Değişken sadece 0 veya 1 değerini alabilir
Integer Variable	Değişken sadece tamsayı değerleri alabilir



Denklemler

- ▶ Denklemler çözülecek veya optimize edilecek modeldeki ilişkileri tanımlar
- ▶ Denklemlerin tanımlanması

EQUATION

birincidenklemadı(ilişkiliküme)

açıklayıcı yazı(isteğe bağlı)

ikincidenklemadı(ilişkiliküme)

açıklayıcı yazı(isteğe bağlı);

```
equations amac, kapasitekisiti, talepkisiti ;
```

EQUATIONS

```
TopZaman          islerin toplam gerçekleştirilme süresi  
makinakisiti(m)   her makinaya sadece bir is atanmalıdır  
iskisiti(i)       bir is sadece bir makinaya atanmalıdır ;
```

Denklemler

► Denklem eşitlikleri

denklemadı(i)lişikiliküme) . .

sol_taraftaki_terimler Eşitlik_türü sağ_taraftaki_terimler;

Kullanılan Sembol	Eşitlik Türü	Koşulun Özelliği
=e=	Eşitlik	soltaraf= sağtaraf
=g=	Büyük ya da eşit	soltaraf ≥ sağtaraf
=l=	Küçük ya da eşit	soltaraf ≤ sağtaraf

```
amac.. z =e= sum((i,j), x(i,j)*c(i,j)) + fa*sum(i, y(i));  
kapasitekisiti(i).. sum(j, x(i,j)) =l= kapasite(i)*y(i);  
talepkisiti(j).. sum(i, x(i,j)) =g= talep(j);
```



Model&Çözüm

- ▶ Modeller GAMS'in çözüm için kullanacağı nesnelerdir

MODEL modeladı açıklayıcı yazı(isteğe bağlı) / model içeriği/;

```
model ulastirma /all/;
```

- ▶ Modeller, model komutu aracılığıyla bir araya getirildikten sonra, solve komutu kullanılarak çözümü istenilen model burada belirtilir

SOLVE modeladı USING modeltipi MAXIMAZING değişkenadı ;

SOLVE modeladı USING modeltipi MINIMAZING değişkenadı ;

```
solve ulastirma using MIP minimizing z;
```



Model&Çözüm

GAMS Model Tipi	Model Tipi Tanımı	Koşullar
LP	Doğrusal Model	Doğrusal olmayan terim ve ayrık değişken içermeyen optimizasyon problemleri
NLP	Doğrusal Olmayan Model	Problem doğrusal olmayan terimler içerebilir ama değişken sürekli olmalıdır
MIP	Karışık Tamsayılı Model	Problem ayrık değişkenler içerebilir ama doğrusal olmayan terim içermemelidir

Bu model tipleri temel modellerdir, problem türüne göre bunların dışında da modeller kullanılabilir. İnternette ve diğer kaynaklarda bunlara kolayca ulaşılabilir



Sabit maliyetli ulařtırma modeli

```
sets i santraller /1, 2, 3/  
     j řehirler /1*4/ ;
```

```
table c(i,j)  
  1   2   3   4  
1  8   6  10   9  
2  9  12  13   7  
3 14   9  16   5 ;
```

```
Scalar fa fabrika acma maliyeti /200000/;  
parameter talep(j) /1 40000, 2 30000, 3 25000, 4 15000 /;  
parameter kapasite(i) / 1 60000, 2 55000, 3 45000 /;
```

```
positive variables x(i,j);  
binary variables y(i);  
variable z;
```

```
equations amac, kapasitekisiti, talepkisiti ;
```

```
amac.. z =e= sum((i,j), x(i,j)*c(i,j)) + fa*sum(i, y(i));  
kapasitekisiti(i).. sum(j, x(i,j)) =l= kapasite(i)*y(i);  
talepkisiti(j).. sum(i, x(i,j)) =g= talep(j);
```

```
model ulastirma /all/;  
solve ulastirma using MIP minimizing z;  
display x.l, y.l;
```

Veri Alma (Küme&Tablo)

- ▶ Problemin çözümü için gerekli tüm verileri GAMS'in içine yazmak yerine, bu verileri saklayan başka bir dosya(excel, notepad) kullanılabilir. Bu sayede hem kod basitleşir hem de var olan verileri tekrar yazmaya gerek kalmaz
- ▶ Importing sets&table from an excel file

	A	B	C	D	E	F
1		m1	m2	m3	m4	
2	f1	8	6	10	9	
3	f2	9	12	13	7	
4	f3	14	9	16	5	
5						

Veri Alma (Küme&Tablo)

SET Kümeadı /

```
$call =xls2gms r=ilkhücre:sonhücre i=dosyaadı.xls o=setKümeadı.inc
```

```
$include setKümeadı.inc
```

```
;/  
sets i fabrikalar /  
$call =xls2gms r=a2:a4 i=data.xls o=seti.inc  
$include seti.inc  
/;
```

- ▶ GAMS sadece düşey verileri okuyabilir, yatay verilerin okunabilmesi için koda ilave bir terim eklenmelidir

SET Kümeadı /

```
$call =xls2gms r= ilkhücre:sonhücre s="," i=dosyaadı.xls o=setKümeadı.inc
```

```
$include set Kümeadı.inc
```

```
;/  
sets      j müşteriler /  
$call =xls2gms r=b1:e1 s="," i=data.xls o=setj.inc  
$include setj.inc  
/;
```

Veri Alma (Küme&Tablo)

- ▶ Tabloları okumak için;

TABLE tabloadı(1.küme,2.küme)

\$call =xls2gms r=hücre(0,0):hücre(n,m) i=dosyaadı.xls o=partabloadı.inc

\$include partabloadı.inc

;

```
TABLE c(i,j)
$call =xls2gms r=a1:e4 i=data.xls o=parc.inc
$include parc.inc
;
```

	A	B	C	D	E	F
1		m1	m2	m3	m4	
2	f1	8	6	10	9	
3	f2	9	12	13	7	
4	f3	14	9	16	5	
5						

```

sets i fabrikalar /
$call =xls2gms r=a2:a4 i=data.xls o=seti.inc
$include seti.inc
/;
sets j müşteriler /
$call =xls2gms r=b1:e1 s="," i=data.xls o=setj.inc
$include setj.inc
/;
TABLE c(i,j)
$call =xls2gms r=a1:e4 i=data.xls o=parc.inc
$include parc.inc
;
scalar fa /200000/;
parameter talep(j) /m1 40000, m2 30000, m3 25000, m4 15000 /;
parameter kapasite(i) / f1 60000, f2 55000, f3 45000 /;

positive variables x(i,j);
binary variables y(i);
variable z;

equations amac, kapasitekisiti, talepkisiti ;

amac.. z =e= sum((i,j), x(i,j)*c(i,j)) + fa*sum(i, y(i));
kapasitekisiti(i).. sum(j, x(i,j)) =l= kapasite(i)*y(i);
talepkisiti(j).. sum(i, x(i,j)) =g= talep(j);

model ulastirma /all/;
solve ulastirma using MIP minimizing z;
display x.l, y.l;


```



Örnek

Küme kapsama problemi

- ▶ Bir şehirde itfaiye merkezlerinin yeri gözden geçirilmektedir. Aşağıdaki şekildeki gibi şehrin onbir adet ilçesi vardır.
- ▶ Bir itfaiye merkezi herhangi bir ilçede kurulabilir. Merkez kendi ilçesindeki ve komşu ilçelerdeki yangınlara müdahale edebilir.
- ▶ Amaç merkez sayısını enazlayarak tüm ili yangına karşı koruyacak bir kapsama alanı kurmaktır.



Çok Dönemli Sermaye Bütçeleme

- Önümüzdeki üç yıllık planlama dönemi için dört projenin değerlendirilmesi yapılacaktır. Her projeye ait beklenen getiriler ile yıllık harcamalar aşağıdaki tabloda gösterilmiştir. Üç yıl boyunca uygulamaya konulacak projeleri belirleyiniz.

		Harcamalar		
Proje	Getiri (NŞD)	1. Yıl	2.Yıl	3.Yıl
1	0.2	0.5	0.3	0.2
2	0.3	1	0.5	0.2
3	0.5	1.5	1.5	0.3
4	0.1	0.1	0.4	0.1
Kullanılabilir sermaye		3.1	2.5	0.4

MS Excel Solver

- ▶ **Sayfanın hazırlanması**
 - ▶ Karar değişkenlerini hücrelere tanımla (her karar değişkeni için ayrı hücre)
 - ▶ Parametre değerleri gir
 - ▶ Amaç fonksiyonunu karar değişkenlerine göre hesaplanmasını içeren bir hücre tanımla
 - ▶ Kısıt sağ taraf ve sol taraf değerlerini ayrı hücrelere olmak üzere tanımla
- ▶ **Solver eklentisinin çalıştırılması**
 - ▶ Amaç fonksiyonu tanımlanan hücreyi işaretle
 - ▶ Min- Max işaretle
 - ▶ Karar değişkenlerini tanımladığın hücreleri işaretle
 - ▶ Kısıtları gir
 - ▶ (gerekli ise) karar değişkeni tipini binary veya integer olarak tanımla
 - ▶ Çözme yöntemi belirle
 - ▶ Çöz



MS Excel Solver

Örnek (Proje Seçimi)

	Yatırımlar(\$)				
	1	2	3	4	5
1. Yıl nakit çıkışı	11	53	5	5	29
2. Yıl nakit çıkışı	3	6	5	1	34
Net şimdiki değer	13	16	16	14	39

1. Yıl kullanılabilir para miktarı 40 br.,
2. Yıl kullanılabilir para miktarı zaman 20 br.

DP modeli:

Karar değişkenleri;

x_i : i . yatırıma para yatırma oranı

$$\text{Maks } z = 13 x_1 + 16 x_2 + 16 x_3 + 14 x_4 + 39 x_5$$

Öyle ki;

$$11 x_1 + 53 x_2 + 5 x_3 + 5 x_4 + 29 x_5 \leq 40$$

$$3 x_1 + 6 x_2 + 5 x_3 + x_4 + 34 x_5 \leq 20$$

$$0 \leq x_i \leq 1 \quad i = 1, \dots, 5$$

Örnek Problem

Sabit maliyetli ulařtırma modeli

- ▶ ATK- Beyaz dört müşterisinin taleplerini karşılamak için üç potansiyel fabrikayı değerlendirmektedir. Her hangi bir fabrikayı açarsa firma 200,000TL'lik yatırım yapması gerekmektedir.
- ▶ Bir fabrikada üretilip, bir müşteriye gönderme birim maliyetleri tabloda verilmiştir.
- ▶ Firmanın talepleri en düşük maliyetle karşılaması için gerekli TP modelini kurunuz.

	Müşteri 1	Müşteri 2	Müşteri 3	Müşteri 4	Kapasite
Fabrika 1	8TL	6TL	10TL	9TL	60,000
Fabrika 2	9TL	12TL	13TL	7TL	55,000
Fabrika 3	14TL	9TL	16TL	5TL	45,000
Talepler	40,000	30,000	25,000	15,000	